Computation without Representation

Thanks for inviting me to Toronto! ωεβ:=http://drorbn.net/to18/

Follows Rozansky [Ro1, Ro2, Ro3] and Overbay [Ov], joint with van der Veen. More at [BV] and at ωεβ/talks.

Abstract. A major part of "quantum topology" is the definition and computation of various knot invariants by carrying out computations in quantum groups. Traditionally these computations are carried out "in a representation", but this is very slow: one has to use tensor powers of these representations, and the dimensions of powers grow exponentially fast.

In my talk, I will describe a direct method for carrying out such computations without having to choose a representation and explain why in many ways the results are better and faster. The two key points we use are a technique for composing infinite-order "perturbed Gaussian" differential operators, and the little-known fact that every semi-simple Lie algebra can be approximated by solvable Lie algebras, where computations are easier.

KiW 43 Abstract (ωεβ/kiw). Whether or not you like the formulas on this page, they describe the strongest truly computable knot invariant we know. (experimental analysis @ωεβ/kiw)

Knotted Candies

ωεβ/kc



The Yang-Baxter Technique. Given an algebra U (typically $\hat{\mathcal{U}}(\mathfrak{g})$ or $\hat{\mathcal{U}}_q(\mathfrak{g})$) and elements

$$R = \sum a_i \otimes b_i \in U \otimes U \quad \text{and} \quad C \in U,$$

form

$$Z = \sum_{i,j,k} C a_i b_j a_k C^2 b_i a_j b_k C.$$

Problem. Extract information from Z.

The Dogma. Use representation theory. In principle finite, but *slow*.

A Knot Theory Portfolio.

- Has operations $\sqcup$, m^{ij}_k, Δ^i_{jk}, S_i.
- All tangloids are generated by $R^{\pm 1}$ and $C^{\pm 1}$ (so "easy" to produce invariants).
- Makes some knot properties ("genus", "ribbon") become "definable".

Tangloids and Operations



A "Quantum Group" Portfolio consists of a vector space U along with maps (and some axioms...)

$$\begin{tikzcd}
\mathbb{Q} = \hat{U}^{\otimes\emptyset} \arrow[r, "C_i"] & \hat{U}^{\otimes\{i\}} \arrow[loop above, "S_i"] \arrow[r, bend left, "\Delta^i_{jk}"] & \hat{U}^{\otimes\{j,k\}} \arrow[l, bend left, "m^{jk}_i"] & \mathbb{Q} = \hat{U}^{\otimes\emptyset} \arrow[l, "R_{jk}"'] \\
\hat{\mathcal{S}}(\emptyset) \arrow[u, "\mathbb{O}_{()}"] \arrow[r, "C_i"] & \hat{\mathcal{S}}(B_i) \arrow[u, "\mathbb{O}_{y_i x_i \dots}"] \arrow[loop below, "S_i"] \arrow[r, bend left, "\Delta^i_{jk}"] & \hat{\mathcal{S}}\left(B_j, B_k\right) \arrow[u, "\mathbb{O}_{y_j x_j \dots \otimes y_k x_k \dots}"] \arrow[l, bend left, "m^{jk}_i"] & \hat{\mathcal{S}}(\emptyset) \arrow[u, "\mathbb{O}_{()}"] \arrow[l, "R_{jk}"']
\end{tikzcd}$$

$\sqcup \to \otimes$

$\sqcup \to \cdot$

Our Way. For certain algebras, work in a homomorphic poly-dimensional "space of formulas".

The (fake) moduli of Lie algebras on V, a quadratic variety in $(V^*)^{\otimes 2} \otimes V$ is on the right. We care about $sl^k_{17} := sl^\epsilon_{17}/(\epsilon^{k+1} = 0)$.



Solvable Approximation. In gl_n, half is enough! Indeed $gl_n \oplus \mathfrak{a}_n = \mathcal{D}(\text{◹}, b, \delta)$:



$b(\text{◹}) = b\colon \text{◹} \otimes \text{◹} \to \text{◹}$

$b(\text{◺}) \rightsquigarrow \delta\colon \text{◹} \to \text{◹} \otimes \text{◹}$

Now define $gl^\epsilon_n := \mathcal{D}(\text{◹}, b, \epsilon\delta)$. Schematically, this is $[\text{◹}, \text{◹}] = \text{◹}$, $[\text{◺}, \text{◺}] = \epsilon\text{◺}$, and $[\text{◹}, \text{◺}] = \text{◺} + \epsilon\text{◹}$. The same process works for all semi-simple Lie algebras, and at $\epsilon^{k+1} = 0$ always yields a solvable Lie algebra.

CU and *QU*. Starting from sl_2, get $CU_\epsilon = \langle y, a, x, t\rangle/([t, -] = 0, [a, y] = -y, [a, x] = x, [x, y] = 2\epsilon a - t)$. Quantize using standard tools (I'm sorry) and get $QU_\epsilon = \langle y, a, x, t\rangle/([t, -] = 0, [a, y] = -y, [a, x] = x, xy - e^{\hbar\epsilon} yx = (1 - Te^{-2\hbar\epsilon a})/\hbar)$.

PBW Bases. The U's we care about always have "Poincaré-Birkhoff-Witt" bases; there is some finite set $B = \{y, x, \dots\}$ of "generators" and isomorphisms $\mathbb{O}_{y,x,\dots}\colon \hat{\mathcal{S}}(B) \to U$ defined by "ordering monomials" to some fixed $y, x, \dots$ order. The quantum group portfolio now becomes a "symmetric algebra" portfolio, or a "power series" portfolio.

Operations are Objects.

★ $$B^* := \{z_i^* = \zeta_i\colon\ z_i \in B\},$$

$$\langle z_i^m, \zeta_i^n\rangle = \delta_{mn} n!,$$

$$\left\langle \prod z_i^{m_i}, \prod \zeta_i^{n_i} \right\rangle = \prod \delta_{m_i n_i} n_i!,$$

in general, for $f \in \mathcal{S}(z_i)$ and $g \in \mathcal{S}(\zeta_i)$,

$$\langle f, g\rangle = f(\partial_{\zeta_i}) g\big|_{\zeta_i=0} = g(\partial_{z_i}) f\big|_{z_i=0}.$$

$$\begin{array}{c}
f \in \mathrm{Hom}_{\mathbb{Q}}(S(B) \to S(B')) \\
\| \\
S(B)^* \otimes S(B') \\
\| \star \\
S(B^*) \otimes S(B') \\
\| \\
S(B^* \sqcup B') \\
\| \\
\tilde{f} \in \mathbb{Q}[\zeta_i, z'_i]
\end{array}$$

The Composition Law. If

$$\mathcal{S}(B) \xrightarrow[\tilde{f} \in \mathbb{Q}[\![\zeta_i, z'_j]\!]]{f} \mathcal{S}(B') \xrightarrow[\tilde{g} \in \mathbb{Q}[\![\zeta'_j, z''_k]\!]]{g} \mathcal{S}(B'')$$

then $$\widetilde{(f /\!\!/ g)} = \widetilde{(g \circ f)} = \left(\tilde{g}\big|_{\zeta'_j \to \partial_{z'_j}} \tilde{f}\right)_{z'_j=0} = \left(\tilde{f}\big|_{z'_j \to \partial_{\zeta'_j}} \tilde{g}\right)_{\zeta'_j=0}:$$



Examples

1. The 1-variable identity map $I\colon \mathcal{S}(z) \to \mathcal{S}(z)$ is given by $\tilde{I}_1 = e^{z\zeta}$ and the n-variable one by $\tilde{I}_n = e^{z_1\zeta_1 + \cdots + z_n\zeta_n}$:



2. The "archetypal multiplication map $m^{ij}_k\colon \mathcal{S}(z_i, z_j) \to \mathcal{S}(z_k)$" has $\tilde{m} = e^{z_k(\zeta_i + \zeta_j)}$.
3. The "archetypal coproduct $\Delta^i_{jk}\colon \mathcal{S}(z_i) \to \mathcal{S}(z_j, z_k)$", given by $z_i \to z_j + z_k$ or $\Delta z = z \otimes 1 + 1 \otimes z$, has $\tilde{\Delta} = e^{(z_j + z_k)\zeta_i}$.
4. R-matrices tend to have terms of the form $e_q^{\hbar y_1 x_2} \in \mathcal{U}_q \otimes \mathcal{U}_q$. The "baby R-matrix" is $\tilde{R} = e^{\hbar yx} \in \mathcal{S}(y, x)$.
5. The "Weyl form of the canonical commutation relations" states that if $[y, x] = tI$ then $e^{\xi x} e^{\eta y} = e^{\eta y} e^{\xi x} e^{-\eta\xi t}$. So with

$$\begin{tikzcd}
\mathcal{S}(y, x) \arrow[loop left, "SW_{xy}"] \arrow[r, shift left, "\mathbb{O}_{xy}"] \arrow[r, shift right, "\mathbb{O}_{yx}"'] & \mathcal{U}(y, x)
\end{tikzcd}$$

we have $\widetilde{SW}_{xy} = e^{\eta y + \xi x - \eta\xi t}$.

The Real Thing. In the algebra QU_ϵ, over $\mathbb{Q}[\![\hbar]\!]$ using the *yaxt* order, $T=\mathbb{e}^{\hbar t}$, $\bar{T}=T^{-1}$, $\mathcal{A}=\mathbb{e}^{\alpha}$, and $\bar{\mathcal{A}}=\mathcal{A}^{-1}$, we have

$$\tilde{R}_{ij}=\mathbb{e}^{\hbar(y_ix_j-t_ia_j)}\left(1+\epsilon\hbar\left(a_ia_j-\hbar^2y_i^2x_j^2/4\right)+O(\epsilon^2)\right)$$

in $\mathcal{S}(B_i,B_j)$, and in $\mathcal{S}(B_1^*,B_2^*,B)$ we have

$$\tilde{m}=\mathbb{e}^{(\alpha_1+\alpha_2)a+\eta_2\xi_1(1-T)/\hbar+(\xi_1\bar{\mathcal{A}}_2+\xi_2)x+(\eta_1+\eta_2\bar{\mathcal{A}}_1)y}\left(1+\epsilon\lambda+O(\epsilon^2)\right),$$

where $\lambda=2a\eta_2\xi_1T+\eta_2^2\xi_1^2\left(3T^2-4T+1\right)/4\hbar-\eta_2\xi_1^2(3T-1)x\bar{\mathcal{A}}_2/2$
$-\eta_2^2\xi_1(3T-1)y\bar{\mathcal{A}}_1/2+\eta_2\xi_1xy\hbar\bar{\mathcal{A}}_1\bar{\mathcal{A}}_2$.

Finally,

$$\tilde{\Delta}=\mathbb{e}^{\tau(t_1+t_1)+\eta(y_1+T_1y_2)+\alpha(a_1+a_2)+\xi(x_1+x_2)}\left(1+O(\epsilon)\right)\in\mathcal{S}(B^*,B_1,B_2),$$

and $\tilde{S}=\mathbb{e}^{-\tau t-\alpha a-\eta\xi(1-\bar{T})\mathcal{A}/\hbar-\bar{T}\eta y\mathcal{A}-\xi x\mathcal{A}}\left(1+O(\epsilon)\right)\in\mathcal{S}(B^*,B)$.

The Zipping Issue. (between unbound and bound lies half-zipped).



Zipping. If $P(\zeta^j,z_i)$ is a polynomial, or whenever otherwise convergent, set $\left\langle P(\zeta^j,z_i)\right\rangle_{(\zeta^j)}=P\left(\partial_{z_j},z_i\right)\Big|_{z_i=0}$. (E.g., if $P=\sum a_{nm}\zeta^nz^m$ then $\langle P\rangle_\zeta=\sum a_{nm}\,\partial_z^nz^m\big|_{z=0}=\sum n!a_{nn}$).

The Zipping / Contraction Theorem. If P has a finite ζ-degree and the y's and the q's are "small" then

$$\left\langle P(z_i,\zeta^j)\mathbb{e}^{c+\eta^iz_i+y_j\zeta^j+q^i_jz_i\zeta^j}\right\rangle_{(\zeta^j)}=\det(\tilde{q})\left\langle P(z_i,\zeta^j)\mathbb{e}^{c+\eta^iz_i}\Big|_{z_i\to\tilde{q}^k_i(z_k+y_k)}\right\rangle_{(\zeta^j)}$$

where $\tilde{q}$ is the inverse matrix of $1-q$: $(\delta^i_j-q^i_j)\tilde{q}^j_k=\delta^i_k$.

Exponential Reservoirs. The true Hilbert hotel is exp! Remove one x from an "exponential reservoir" of x's and you are left with the same exponential reservoir:

$$\mathbb{e}^x=\left[\ldots+\frac{xxxxx}{120}+\ldots\right]\xrightarrow{\partial_x}\left[\ldots+\frac{x\cancel{x}xxx}{120}+\ldots\right]=(\mathbb{e}^x)'=\mathbb{e}^x,$$

and if you let each element choose left or right, you get twice the same reservoir:

$$\mathbb{e}^x\xrightarrow{x\to x_l+x_r}\mathbb{e}^{x_l+x_r}=\mathbb{e}^{x_l}\mathbb{e}^{x_r}.$$



A Graphical Proof. Glue top to bottom on the right, in all possible ways. Several scenarios occur:

1. Start at A, go through the q-machine $k\geq 0$ times, stop at B. Get $\left\langle P\left(\sum_{k\geq 0}q^kz,\zeta\right)\right\rangle=\langle P(\tilde{q}z,\zeta)\rangle$.
2. Loop through the q-machine and swallow your own tail. Get $\exp\left(\sum q^k/k\right)=\exp(-\log(1-q))=\tilde{q}$.
3. …

By the reservoir splitting principle, these scenarios contribute multiplicatively. □

Implementation. ωεβ/Zip

```
𝔼 /: Zip_{ζs_List}@𝔼[Q_, P_] :=      (* 𝔼[Q,P] means e^Q P *)
  Module[{ζ, z, zs, c, ys, ηs, qt, zrule, Q1, Q2},
   zs = Table[ζ*, {ζ, ζs}];
   c = Q /. Alternatives @@ (ζs ⋃ zs) → 0;
   ys = Table[∂_ζ (Q /. Alternatives @@ zs → 0), {ζ, ζs}];
   ηs = Table[∂_z (Q /. Alternatives @@ ζs → 0), {z, zs}];
   qt = Inverse@Table[Kδ_{z,ζ*} - ∂_{z,ζ}Q, {ζ, ζs}, {z, zs}];
   zrule = Thread[zs → qt.(zs + ys)];
   Q1 = c + ηs.zs /. zrule; Q2 = Q1 /. Alternatives @@ zs → 0;
   Simplify /@ 𝔼[Q2, Det[qt] e^{-Q2} Zip_{ζs}[e^{Q1} (P /. zrule)]] ];
```

Real Zipping is a minor mess, and is done in two phases:

	τa-phase		ξy-phase	
ζ-like variables	τ	a	ξ	y
z-like variables	t	α	x	η

Already at $\epsilon=0$ we get the best known formulas for the Alexander polynomial!

Generic Docility. A "docile perturbed Gaussian" in the variables $(z_i)_{i\in S}$ over the ring R is an expression of the form

$$\mathbb{e}^{q^{ij}z_iz_j}P=\mathbb{e}^{q^{ij}z_iz_j}\left(\sum_{k\geq 0}\epsilon^kP_k\right),$$

where all coefficients are in R and where P is a "docile series": $\deg P_k\leq 4k$.

Our Docility. In the case of QU_ϵ, all invariants and operations are of the form $\mathbb{e}^{L+Q}P$, where

- L is a quadratic of the form $\sum l_{z\zeta}z\zeta$, where z runs over $\{t_i,\alpha_i\}_{i\in S}$ and ζ over $\{\tau_i,a_i\}_{i\in S}$, with integer coefficients $l_{z\zeta}$.
- Q is a quadratic of the form $\sum q_{z\zeta}z\zeta$, where z runs over $\{x_i,\eta_i\}_{i\in S}$ and ζ over $\{\xi_i,y_i\}_{i\in S}$, with coefficients $q_{z\zeta}$ in the ring R_S of rational functions in $\{T_i,\mathcal{A}_i\}_{i\in S}$.
- P is a docile power series in $\{y_i,a_i,x_i,\eta_i,\xi_i\}_{i\in S}$ with coefficients in R_S, and where $\deg(y_i,a_i,x_i,\eta_i,\xi_i)=(1,2,1,1,1)$.

Docililty Matters! The rank of the space of docile series to ϵ^k is polynomial in the number of variables $|S|$. !!!!!

- At $\epsilon^2=0$ we get the Rozansky-Overbay [Ro1, Ro2, Ro3, Ov] invariant, which is stronger than HOMFLY-PT polynomial and Khovanov homology taken together!
- In general, get "higher diagonals in the Melvin-Morton-Rozansky expansion of the coloured Jones polynomial" [MM, BNG], but why spoil something good?

References.

[BNG] D. Bar-Natan and S. Garoufalidis, *On the Melvin-Morton-Rozansky conjecture,* Invent. Math. **125** (1996) 103–133.

[BV] D. Bar-Natan and R. van der Veen, *A Polynomial Time Knot Polynomial,* arXiv:1708.04853.

[Fa] L. Faddeev, *Modular Double of a Quantum Group,* arXiv:math/9912078.

[GR] S. Garoufalidis and L. Rozansky, *The Loop Exapnsion of the Kontsevich Integral, the Null-Move, and S-Equivalence,* arXiv:math.GT/0003187.

[MM] P. M. Melvin and H. R. Morton, *The coloured Jones function,* Commun. Math. Phys. **169** (1995) 501–520.

[Ov] A. Overbay, *Perturbative Expansion of the Colored Jones Polynomial,* University of North Carolina PhD thesis, ωεβ/Ov.

[Qu] C. Quesne, *Jackson's q-Exponential as the Exponential of a Series,* arXiv: math-ph/0305003.

[Ro1] L. Rozansky, *A contribution of the trivial flat connection to the Jones polynomial and Witten's invariant of 3d manifolds, I,* Comm. Math. Phys. **175-2** (1996) 275–296, arXiv:hep-th/9401061.

[Ro2] L. Rozansky, *The Universal R-Matrix, Burau Representation and the Melvin-Morton Expansion of the Colored Jones Polynomial,* Adv. Math. **134-1** (1998) 1–31, arXiv:q-alg/9604005.

[Ro3] L. Rozansky, *A Universal U(1)-RCC Invariant of Links and Rationality Conjecture,* arXiv:math/0201139.

[Za] D. Zagier, *The Dilogarithm Function,* in Cartier, Moussa, Julia, and Vanhove (eds) *Frontiers in Number Theory, Physics, and Geometry II.* Springer, Berlin, Heidelberg, and ωεβ/Za.







"God created the knots, all else in topology is the work of mortals."

Leopold Kronecker (modified)

www.katlas.org

The Algebras H and H^*. Let $q = \mathbb{e}^{\hbar\epsilon\gamma}$ and set $H = \langle a, x\rangle/([a,x]=\gamma x)$ with

$$A = \mathbb{e}^{-\hbar\epsilon a}, \quad xA = qAx, \quad S_H(a,A,x) = (-a, A^{-1}, -A^{-1}x),$$
$$\Delta_H(a,A,x) = (a_1+a_2, A_1A_2, x_1+A_1x_2)$$

and dual $H^* = \langle b, y\rangle/([b,y] = -\epsilon y)$ with

$$B = \mathbb{e}^{-\hbar\gamma b}, \quad By = qyB, \quad S_{H^*}(b,B,y) = (-b, B^{-1}, -yB^{-1}),$$
$$\Delta_{H^*}(b,B,y) = (b_1+b_2, B_1B_2, y_1B_2+y_2).$$

Pairing by $(a,x)^* = (b,y)$ ($\Rightarrow \langle B, A\rangle = q$) making $\langle y^l b^i, a^j x^k\rangle = \delta_{ij}\delta_{kl} j![k]_q!$ so $R = \sum \frac{y^k b^j \otimes a^j x^k}{j![k]_q!}$.

The Algebra QU. Using the Drinfel'd double procedure, $QU_{\gamma,\epsilon} := H^{*cop} \otimes H$ with $(\phi f)(\psi g) = \langle \psi_1 S^{-1} f_3\rangle\langle\psi_3, f_1\rangle(\phi\psi_2)(f_2 g)$ and

$$S(y,b,a,x) = (-B^{-1}y, -b, -a, -A^{-1}x),$$
$$\Delta(y,b,a,x) = (y_1+y_2B_1, b_1+b_2, a_1+a_2, x_1+A_1x_2).$$

Note also that $t := \epsilon a - \gamma b$ is central and can replace b, and set $QU = QU_\epsilon = QU_{1,\epsilon}$.

The 2D Lie Algebra. One may show* that if $[a,x] = \gamma x$ then $\mathbb{e}^{\xi x}\mathbb{e}^{\alpha a} = \mathbb{e}^{\alpha a}\mathbb{e}^{\mathbb{e}^{-\gamma\alpha}\xi x}$. Ergo with

$$\begin{tikzcd} \mathcal{S}(a,x) \arrow[loop left, "SW_{ax}"] \arrow[r, shift left, "\mathbb{O}_{ax}"] \arrow[r, shift right, "\mathbb{O}_{xa}"'] & \mathcal{U}(a,x) \end{tikzcd}$$

we have $\widetilde{SW}_{ax} = \mathbb{e}^{\alpha a + \mathbb{e}^{-\gamma\alpha}\xi x}$.

$*$ Indeed $xa = (a-\gamma)x$ thus $xa^n = (a-\gamma)^n x$ thus $x\mathbb{e}^{\alpha a} = \mathbb{e}^{\alpha(a-\gamma)}x = \mathbb{e}^{-\gamma\alpha}\mathbb{e}^{\alpha a}x$ thus $x^n\mathbb{e}^{\alpha a} = \mathbb{e}^{\alpha a}(\mathbb{e}^{-\gamma\alpha})^n x^n$ thus $\mathbb{e}^{\xi x}\mathbb{e}^{\alpha a} = \mathbb{e}^{\alpha a}\mathbb{e}^{\mathbb{e}^{-\gamma\alpha}\xi x}$.

Faddeev's Formula (In as much as we can tell, first appeared without proof in Faddeev [Fa], rediscovered and proven in Quesne [Qu], and again with easier proof, in Zagier [Za]). With $[n]_q := \frac{q^n-1}{q-1}$, with $[n]_q! := [1]_q[2]_q\cdots[n]_q$ and with $\mathbb{e}_q^x := \sum_{n\geq 0}\frac{x^n}{[n]_q!}$, we have

$$\log \mathbb{e}_q^x = \sum_{k\geq 1}\frac{(1-q)^k x^k}{k(1-q^k)} = x + \frac{(1-q)^2x^2}{2(1-q^2)} + \ldots.$$

Proof. We have that $\mathbb{e}_q^x = \frac{\mathbb{e}_q^{qx} - \mathbb{e}_q^x}{qx - x}$ ("the q-derivative of $\mathbb{e}_q^x$ is itself"), and hence $\mathbb{e}_q^{qx} = (1+(1-q)x)\mathbb{e}_q^x$, and

$$\log \mathbb{e}_q^{qx} = \log(1+(1-q)x) + \log \mathbb{e}_q^x.$$

Writing $\log \mathbb{e}_q^x = \sum_{k\geq 1} a_k x^k$ and comparing powers of x, we get $q^k a_k = -(1-q)^k/k + a_k$, or $a_k = \frac{(1-q)^k}{k(1-q^k)}$. □

A Full Implementation. ωεβ/Full

Utilities

```
CF[sd_SeriesData] := MapAt[CF, sd, 3];
CF[ℰ_] := ExpandDenominator@ExpandNumerator@Together[
    Expand[ℰ] //. e^x_ e^y_ :> e^(x+y) /. e^x_ :> e^CF[x]];
```

```
Kδ /: Kδ_{i_,j_} := If[i === j, 1, 0];
```

```
𝔼 /: 𝔼[L1_, Q1_, P1_] ≡ 𝔼[L2_, Q2_, P2_] :=
  CF[L1 == L2] ∧ CF[Q1 == Q2] ∧ CF[Normal[P1 - P2] == 0];
𝔼 /: 𝔼[L1_, Q1_, P1_] 𝔼[L2_, Q2_, P2_] :=
  𝔼[L1 + L2, Q1 + Q2, P1 * P2];
𝔼[L_, Q_, P_]_{$k_} := 𝔼[L, Q, Series[Normal@P, {ϵ, 0, $k}]];
```

Zip and Bind

```
{t*, b*, y*, a*, x*, z*} = {τ, β, η, α, ξ, ζ};
{τ*, β*, η*, α*, ξ*, ζ*} = {t, b, y, a, x, z};
(u_i_)* := (u*)_i;
```

```
collect[sd_SeriesData, ζ_] :=
  MapAt[collect[#, ζ] &, sd, 3];
collect[ℰ_, ζ_] := Collect[ℰ, ζ];
Zip_{}[P_] := P; Zip_{ζ_,ζs___}[P_] :=
 (collect[P // Zip_{ζs}, ζ] /. f_. ζ^d_. :> ∂_{ζ*,d} f) /. ζ* → 0
```

```
QZip_{ζs_List}@𝔼[L_, Q_, P_] :=
  Module[{ζ, z, zs, c, ys, ηs, qt, zrule, Q1, Q2},
   zs = Table[ζ*, {ζ, ζs}];
   c = Q /. Alternatives @@ (ζs ⋃ zs) → 0;
   ys = Table[∂_ζ (Q /. Alternatives @@ zs → 0), {ζ, ζs}];
   ηs = Table[∂_z (Q /. Alternatives @@ ζs → 0), {z, zs}];
   qt = Inverse@Table[Kδ_{z,ζ*} - ∂_{z,ζ}Q, {ζ, ζs}, {z, zs}];
   zrule = Thread[zs → qt.(zs + ys)];
   Q2 = (Q1 = c + ηs.zs /. zrule) /. Alternatives @@ zs → 0;
   CF /@ 𝔼[L, Q2, Det[qt] e^(-Q2) Zip_{ζs}[e^Q1 (P /. zrule)]]];
```

```
U2l = {B_i_^p_. → e^(-p ħ γ b_i), B^p_. → e^(-p ħ γ b), T_i_^p_. → e^(p ħ t_i),
   T^p_. → e^(p ħ t), 𝒜_i_^p_. → e^(p γ α_i), 𝒜^p_. → e^(p γ α)};
l2U = {e^(c_. b_i_ + d_.) :> B_i^(-c/(ħ γ)) e^d, e^(c_. b + d_.) :> B^(-c/(ħ γ)) e^d,
   e^(c_. t_i_ + d_.) :> T_i^(c/ħ) e^d, e^(c_. t + d_.) :> T^(c/ħ) e^d,
   e^(c_. α_i_ + d_.) :> 𝒜_i^(c/γ) e^d, e^(c_. α + d_.) :> 𝒜^(c/γ) e^d,
   e^ℰ_ :> e^Expand@ℰ};
```

```
LZip_{ζs_List}@𝔼[L_, Q_, P_] :=
  Module[{ζ, z, zs, c, ys, ηs, lt, zrule, L1, L2, Q1, Q2},
   zs = Table[ζ*, {ζ, ζs}];
   c = L /. Alternatives @@ (ζs ⋃ zs) → 0;
   ys = Table[∂_ζ (L /. Alternatives @@ zs → 0), {ζ, ζs}];
   ηs = Table[∂_z (L /. Alternatives @@ ζs → 0), {z, zs}];
   lt = Inverse@Table[Kδ_{z,ζ*} - ∂_{z,ζ}L, {ζ, ζs}, {z, zs}];
   zrule = Thread[zs → lt.(zs + ys)];
   L2 = (L1 = c + ηs.zs /. zrule) /. Alternatives @@ zs → 0;
   Q2 = (Q1 = Q /. U2l /. zrule) /. Alternatives @@ zs → 0;
   CF /@ 𝔼[L2, Q2, Det[lt] e^(-L2-Q2)
       Zip_{ζs}[e^(L1+Q1) (P /. U2l /. zrule)]] //. l2U];
```

```
B_{}[L_, R_] := L R;
B_{is__}[L_𝔼, R_𝔼] := Module[{n}, Times[
      L /. Table[(v : b | B | t | T | a | x | y)_i → v_{n@i}, {i, {is}}],
      R /. Table[(v : β | τ | α | 𝒜 | ξ | η)_i → v_{n@i}, {i, {is}}]
    ] // LZip_{Join@@Table[{β_{n@i},τ_{n@i},a_{n@i}},{i,{is}}]} //
   QZip_{Join@@Table[{ξ_{n@i},y_{n@i}},{i,{is}}]} ];
B_{is___}[L_, R_] := B_{{is}}[L, R];
```

𝔼 morphisms with domain and range.

```
B_{is_List}[𝔼_{d1_→r1_}[L1_, Q1_, P1_], 𝔼_{d2_→r2_}[L2_, Q2_, P2_]] :=
  𝔼_{(d1⋃Complement[d2,is])→(r2⋃Complement[r1,is])} @@
   B_{is}[𝔼[L1, Q1, P1], 𝔼[L2, Q2, P2]];
𝔼_{d1_→r1_}[L1_, Q1_, P1_] // 𝔼_{d2_→r2_}[L2_, Q2_, P2_] :=
  B_{r1⋂d2}[𝔼_{d1→r1}[L1, Q1, P1], 𝔼_{d2→r2}[L2, Q2, P2]];
𝔼_{d1_→r1_}[L1_, Q1_, P1_] ≡ 𝔼_{d2_→r2_}[L2_, Q2_, P2_] ^:=
  (d1 == d2) ∧ (r1 == r2) ∧ (𝔼[L1, Q1, P1] ≡ 𝔼[L2, Q2, P2]);
𝔼_{d1_→r1_}[L1_, Q1_, P1_] 𝔼_{d2_→r2_}[L2_, Q2_, P2_] ^:=
  𝔼_{(d1⋃d2)→(r1⋃r2)} @@ (𝔼[L1, Q1, P1] 𝔼[L2, Q2, P2]);
𝔼_{d_→r_}[L_, Q_, P_]_{$k_} := 𝔼_{d→r} @@ 𝔼[L, Q, P]_{$k};
𝔼_[ℰ___][i_] := {ℰ}[[i]];
```

"Define" code

```
SetAttributes[Define, HoldAll];
Define[def_, defs__] := (Define[def]; Define[defs];);
```

```
Define[op_is__ = ℰ_] :=
 Module[{SD, ii, jj, kk, isp, nis, nisp, sis},
  Block[{i, j, k},
   ReleaseHold[Hold[
      SD[op_{nisp,$k_Integer}, Block[{i, j, k}, op_{isp,$k} = ℰ;
         op_{nis,$k}]];
      SD[op_isp, op_{{is},$k}]; SD[op_sis__, op_{{sis}}];
     ] /. {SD → SetDelayed,
      isp → {is} /. {i → i_, j → j_, k → k_},
      nis → {is} /. {i → ii, j → jj, k → kk},
      nisp → {is} /. {i → ii_, j → jj_, k → kk_}
     }] ]]
```

The Fundamental Tensors

```
Define[am_{i,j→k} = 𝔼_{{i,j}→{k}}[(α_i + α_j) a_k, (e^{-γ α_j} ξ_i + ξ_j) x_k, 1]_$k,
 bm_{i,j→k} = 𝔼_{{i,j}→{k}}[(β_i + β_j) b_k, (η_i + η_j) y_k, e^{(e^{-∈ β_i} - 1) η_j y_k}]_$k]

Define[R_{i,j} =
 𝔼_{{}→{i,j}}[ħ a_j b_i, ħ x_j y_i, e^(Sum_{k=2}^{$k+1} ((1 - e^{γ∈ħ})^k (ħ y_i x_j)^k) / (k (1 - e^{k γ∈ħ})))]_$k]

Define[R̄_{i,j} = 𝔼_{{}→{i,j}}[-ħ a_j b_i, -ħ x_j y_i / B_i,
   1 + If[$k == 0, 0, (R̄_{{i,j},$k-1})_$k[3] -
     (((R̄_{{i,j},0})_$k R_{1,2} (R̄_{{3,4},$k-1})_$k) // (bm_{i,1→i} am_{j,2→j}) //
       (bm_{i,3→i} am_{j,4→j}))[3]]],
 P_{i,j} = 𝔼_{{i,j}→{}}[β_i α_j / ħ, η_i ξ_j / ħ,
   1 + If[$k == 0, 0, (P_{{i,j},$k-1})_$k[3] -
     (R_{1,2} // ((P_{{1,j},0})_$k (P_{{i,2},$k-1})_$k))[3]]]]

Define[aS_j = R̄_{i,j} ~ B_i ~ P_{i,j},
 aS̄_i = 𝔼_{{i}→{i}}[-a_i α_i, -x_i 𝒜_i ξ_i,
   1 + If[$k == 0, 0, (aS̄_{{i},$k-1})_$k[3] -
     ((aS̄_{{i},0})_$k ~ B_i ~ aS_i ~ B_i ~ (aS̄_{{i},$k-1})_$k)[3]]]]

Define[bS_i = R_{i,1} ~ B_1 ~ aS_1 ~ B_1 ~ P_{i,1},
 bS̄_i = R_{i,1} ~ B_1 ~ aS̄_1 ~ B_1 ~ P_{i,1},
 aΔ_{i→j,k} = (R_{1,j} R_{2,k}) // bm_{1,2→3} // P_{3,i},
 bΔ_{i→j,k} = (R_{j,1} R_{k,2}) // am_{1,2→3} // P_{i,3}]
```

```
Define[
 dm_{i,j→k} =
  (𝔼_{{i,j}→{i,j}}[β_i b_i + α_j a_j, η_i y_i + ξ_j x_j, 1]
     (aΔ_{i→1,2} // aΔ_{2→2,3} // aS̄_3) (bΔ_{j→-1,-2} // bΔ_{-2→-2,-3})) //
   (P_{-1,3} P_{-3,1} am_{2,j→k} bm_{i,-2→k}),
 dS_i = 𝔼_{{i}→{1,2}}[β_i b_1 + α_i a_2, η_i y_1 + ξ_i x_2, 1] // (bS̄_1 aS_2) //
   dm_{2,1→i},
 dΔ_{i→j,k} = (bΔ_{i→3,1} aΔ_{i→2,4}) // (dm_{3,4→k} dm_{1,2→j})]

Define[C_i = 𝔼_{{}→{i}}[0, 0, B_i^{1/2} e^{-ħ∈a_i/2}]_$k,
 C̄_i = 𝔼_{{}→{i}}[0, 0, B_i^{-1/2} e^{ħ∈a_i/2}]_$k,
 Kink_i = (R_{1,3} C̄_2) // dm_{1,2→1} // dm_{1,3→i},
 Kink̄_i = (R̄_{1,3} C_2) // dm_{1,2→1} // dm_{1,3→i}]

Define[
 b2t_i = 𝔼_{{i}→{i}}[α_i a_i - β_i t_i / γ, ξ_i x_i + η_i y_i, e^{∈ β_i a_i / γ}]_$k,
 t2b_i = 𝔼_{{i}→{i}}[α_i a_i - τ_i γ b_i, ξ_i x_i + η_i y_i, e^{∈ τ_i a_i}]_$k]

Define[kR_{i,j} = R_{i,j} // (b2t_i b2t_j) /. t_{i|j} → t,
 kR̄_{i,j} = R̄_{i,j} // (b2t_i b2t_j) /. {t_{i|j} → t, T_{i|j} → T},
 km_{i,j→k} = (t2b_i t2b_j) // dm_{i,j→k} //
   b2t_k /. {t_k → t, T_k → T, τ_{i|j} → 0},
 kC_i = C_i // b2t_i /. T_i → T, kC̄_i = C̄_i // b2t_i /. T_i → T,
 kKink_i = Kink_i // b2t_i /. {t_i → t, T_i → T},
 kKink̄_i = Kink̄_i // b2t_i /. {t_i → t, T_i → T}]
```

The Trefoil

```
$k = 2; Z = kR_{1,5} kR_{6,2} kR_{3,7} kC̄_4 kKink̄_8 kKink̄_9 kKink̄_10;
Do[Z = Z ~ B_{1,r} ~ km_{1,r→1}, {r, 2, 10}];
Simplify /@ Z /. v_1 :→ v
```

$$\mathbb{E}_{\{\}\to\{1\}}\Big[0,\,0,\,\frac{T}{1-T+T^2}+\frac{1}{(1-T+T^2)^3}T\hbar\left(2a\left(-1+T-T^3+T^4\right)+T\left(-1+2T-3T^2+2T^3\right)\gamma-2\left(1+T^3\right)xy\gamma\hbar\right)\epsilon+\frac{1}{2\left(1-T+T^2\right)^5}T\hbar^2\Big(4a^2\left(1-T+T^2\right)^2\left(1+T-6T^2+T^3+T^4\right)+4a\left(1-T+T^2\right)\gamma\left(T\left(2-5T+8T^2-7T^3-2T^4+2T^5\right)-2\left(-1-2T+5T^2-4T^3+T^4+2T^5\right)xy\hbar\right)+\gamma^2\left(T\left(1-2T+4T^2-2T^3+6T^5-11T^6+4T^7\right)+4\left(-1+2T+T^3+T^4+2T^6-T^7\right)xy\hbar+6\left(1-T+T^2\right)^2\left(1+3T+T^2\right)x^2y^2\hbar^2\right)\Big)\epsilon^2+O[\epsilon]^3\Big]$$

diagram	n_k^t	Alexander's ω^+ Today's ρ_1^+	genus / ribbon unknotting # / amphi?	diagram	n_k^t	Alexander's ω^+ Today's ρ_1^+	genus / ribbon unknotting # / amphi?	diagram	n_k^t	Alexander's ω^+ Today's ρ_1^+	genus / ribbon unknotting # / amphi?
	0_1^a	1 0	0 / ✔ 0 / ✔		3_1^a	$t-1$ t	1 / ✘ 1 / ✘		4_1^a	$3-t$ 0	1 / ✘ 1 / ✔
	5_1^a	t^2-t+1 $2t^3+3t$	2 / ✘ 2 / ✘		5_2^a	$2t-3$ $5t-4$	1 / ✘ 1 / ✘		6_1^a	$5-2t$ $t-4$	1 / ✔ 1 / ✘
	6_2^a	$-t^2+3t-3$ t^3-4t^2+4t-4	2 / ✘ 1 / ✘		6_3^a	t^2-3t+5 0	2 / ✘ 1 / ✔		7_1^a	t^3-t^2+t-1 $3t^5+5t^3+6t$	3 / ✘ 3 / ✘
	7_2^a	$3t-5$ $14t-16$	1 / ✘ 1 / ✘		7_3^a	$2t^2-3t+3$ $-9t^3+8t^2-16t+12$	2 / ✘ 2 / ✘		7_4^a	$4t-7$ $32-24t$	1 / ✘ 2 / ✘
	7_5^a	$2t^2-4t+5$ $9t^3-16t^2+29t-28$	2 / ✘ 2 / ✘		7_6^a	$-t^2+5t-7$ $t^3-8t^2+19t-20$	2 / ✘ 1 / ✘		7_7^a	t^2-5t+9 $8-3t$	2 / ✘ 1 / ✘
	8_1^a	$7-3t$ $5t-16$	1 / ✘ 1 / ✘		8_2^a	$-t^3+3t^2-3t+3$ $2t^5-8t^4+10t^3-12t^2+13t-12$	3 / ✘ 2 / ✘		8_3^a	$9-4t$ 0	1 / ✘ 2 / ✔
	8_4^a	$-2t^2+5t-5$ $3t^3-8t^2+6t-4$	2 / ✘ 2 / ✘		8_5^a	$-t^3+3t^2-4t+5$ $-2t^5+8t^4-13t^3+20t^2-22t+24$	3 / ✘ 2 / ✘		8_6^a	$-2t^2+6t-7$ $5t^3-20t^2+28t-32$	2 / ✘ 2 / ✘
	8_7^a	t^3-3t^2+5t-5 $-t^5+4t^4-10t^3+12t^2-13t+12$	3 / ✘ 1 / ✘		8_8^a	$2t^2-6t+9$ $-t^3+4t^2-12t+16$	2 / ✔ 2 / ✘		8_9^a	$-t^3+3t^2-5t+7$ 0	3 / ✔ 1 / ✔
	8_{10}^a	t^3-3t^2+6t-7 $-t^5+4t^4-11t^3+16t^2-21t+20$	3 / ✘ 2 / ✘		8_{11}^a	$-2t^2+7t-9$ $5t^3-24t^2+39t-44$	2 / ✘ 1 / ✘		8_{12}^a	$t^2-7t+13$ 0	2 / ✘ 2 / ✔
	8_{13}^a	$2t^2-7t+11$ $-t^3+4t^2-14t+20$	2 / ✘ 1 / ✘		8_{14}^a	$-2t^2+8t-11$ $5t^3-28t^2+57t-68$	2 / ✘ 1 / ✘		8_{15}^a	$3t^2-8t+11$ $21t^3-64t^2+120t-140$	2 / ✘ 2 / ✘
	8_{16}^a	t^3-4t^2+8t-9 $t^5-6t^4+17t^3-28t^2+35t-36$	3 / ✘ 2 / ✘		8_{17}^a	$-t^3+4t^2-8t+11$ 0	3 / ✘ 1 / ✔		8_{18}^a	$-t^3+5t^2-10t+13$ 0	3 / ✘ 2 / ✔
	8_{19}^n	t^3-t^2+1 $-3t^5-4t^2-3t$	3 / ✘ 3 / ✘		8_{20}^n	t^2-2t+3 $4t-4$	2 / ✔ 1 / ✘		8_{21}^n	$-t^2+4t-5$ $t^3-8t^2+16t-20$	2 / ✘ 1 / ✘